

G. CORALLO

Dipartimento Sostenibilità dei Sistemi Produttivi e Territoriali
Servizio Tecnico Strategico
Centro Ricerche Casaccia, Roma

**REALIZZAZIONE DI UN NUOVO SISTEMA
DI ACQUISIZIONE DATI
PER L' IMPIANTO DI SOLAR COOLING
DELL' EDIFICIO F96 "SERRA A CONTENIMENTO"
DEL CENTRO CASACCIA**

RT/2016/29/ENEA



AGENZIA NAZIONALE PER LE NUOVE TECNOLOGIE,
L'ENERGIA E LO SVILUPPO ECONOMICO SOSTENIBILE

G. CORALLO

Dipartimento Sostenibilità dei Sistemi Produttivi e Territoriali
Servizio Tecnico Strategico
Centro Ricerche Casaccia, Roma

REALIZZAZIONE DI UN NUOVO SISTEMA
DI ACQUISIZIONE DATI
PER L' IMPIANTO DI SOLAR COOLING
DELL' EDIFICIO F96 "SERRA A CONTENIMENTO"
DEL CENTRO CASACCIA

RT/2016/29/ENEA



AGENZIA NAZIONALE PER LE NUOVE TECNOLOGIE,
L'ENERGIA E LO SVILUPPO ECONOMICO SOSTENIBILE

I rapporti tecnici sono scaricabili in formato pdf dal sito web ENEA alla pagina <http://www.enea.it/it/produzione-scientifica/rapporti-tecnici>

I contenuti tecnico-scientifici dei rapporti tecnici dell'ENEA rispecchiano l'opinione degli autori e non necessariamente quella dell'Agenzia

The technical and scientific contents of these reports express the opinion of the authors but not necessarily the opinion of ENEA.

REALIZZAZIONE DI UN NUOVO SISTEMA DI ACQUISIZIONE DATI PER L' IMPIANTO DI SOLAR COOLING DELL' EDIFICIO F96 "SERRA A CONTENIMENTO" DEL CENTRO CASACCIA

G. Corallo

Riassunto

La serra a contenimento del centro ENEA della Casaccia è stata dotata nel 2012 di un impianto di solar cooling sia a scopo dimostrativo che di contenimento dei consumi energetici del sistema di condizionamento dell'aria.

Tale impianto ha richiesto il rifacimento del sistema di acquisizione dati originale per il guasto della maggior parte dei sensori installati.

Il nuovo sistema di acquisizione dati, molto più completo del precedente, è stato realizzato a costo zero, utilizzando sensori, data logger e PC in disuso, reperiti presso i laboratori del Centro Casaccia e adottando il software HT Basic 9.1.

Parole chiave: Solar Cooling, Acquisizione Dati, Chiller ad Assorbimento, Pannelli Solari, HT Basic.

Abstract

The Containing Greenhouse of ENEA's Casaccia Centre was equipped in 2012 with a Solar Cooling plant, with demonstration purposes and in order to limit the electricity consumption of the air conditioning system.

The plant required the remake of the data acquisition system due to the failure of most the installed sensors.

The new data acquisition system, much more complete of the previous one, has been realized at near no cost, adopting abandoned sensors, data logger and PC, found near the laboratories of the Casaccia Centre and the HT Basic 9.1 Software.

Keywords: Solar Cooling, Data Acquisition, Absorption Chiller, Solar Panels, HT Basic.

INDICE

Introduzione	7
1 - Schema dell'impianto	7
2 - Sensori utilizzati	8
2.1 Sensori Pt100, correnti liquide	8
2.2 Sonda NTC, temperatura dell'aria esterna	10
2.3 Misuratori di flusso ad induzione magnetica, correnti liquide	10
3 - Acquisizione dati	11
3.1 Programma di acquisizione dati	14
Conclusioni	16
Ringraziamenti	16
Bibliografia	17
Appendice A: listato del programma	18

Introduzione

L'impianto di Solar Cooling dell' edificio F 96, "Serra a Contenimento", realizzato nel 2010 con fondi dell' Accordo di Programma ENEA-MISE [1 ; 2], era originariamente dotato di n° 6 misuratori di portata ad induzione magnetica e da 10 misuratori remoti di temperatura sulle correnti principali dell' impianto con trasmissione WiFi.

In realtà, solamente i misuratori di temperatura venivano acquisiti in tempo reale tramite una centralina Wi Fi dedicata e dotata anche di interfaccia WEB.

Tale soluzione temporanea, oltre a non permettere il calcolo istantaneo delle prestazioni dell' impianto, richiedeva continui cambi delle pile di alimentazione dei sensori remoti di temperatura.

A causa della non elevata qualità costruttiva tali sensori si sono man mano deteriorati, rendendo necessaria la sostituzione.

Per ovviare alle carenze del sistema sin qui utilizzato, si è quindi deciso di realizzare un nuovo sistema di acquisizione dati, molto più completo, basato interamente su materiale in disuso ed obsoleto ma ancora perfettamente funzionante, reperito presso i laboratori del centro Casaccia.

1-Schema dell' impianto

L'impianto di solar cooling installato presso la serra a contenimento dell' unità SSPT-BIOAG-BIOTEC, è basato su un chiller ad assorbimento Yazaki alimentato ad acqua calda: tale macchina produce acqua refrigerata a circa 7°C, da utilizzare nei sistemi di distribuzione all' interno della serra, mediante apporto di energia termica sotto forma di acqua calda, ad una temperatura compresa tra 75 e 95 °C.

Con riferimento allo schema semplificato di Fig.1, tale impianto è composto da un primo sottoinsieme solare, che ha lo scopo di preparare l' acqua calda di alimentazione

Il sottoinsieme solare è a sua volta costituito dai **Pannelli Solari**, dalla pompa di circolazione **P1**, dal **serbatoio di accumulo** acqua calda, dal **raffreddatore di emergenza**, dalla valvola motorizzata a tre vie **VM**, dalla **caldaia di integrazione** con pompa di circolazione **P2**, dallo **scambiatore di calore** e da una centralina solare di gestione.

La centralina solare ha il compito di accendere con logica ON-OFF la pompa P1, del tipo a portata fissa, se la temperatura in uscita dal campo solare è superiore a quella del fondo del serbatoio AC a meno del valore di isteresi prefissato.

L' acqua calda preparata dal sottoinsieme solare, a temperature comprese tra 70 e 95 °C, viene poi inviata tramite la Pompa **P3** al generatore di **Macchina ad Assorbimento**, per il tramite dello scambiatore di calore connesso alla Caldaia.

Il chiller ad assorbimento per poter funzionare deve smaltire calore per il tramite della **torre evaporativa**, alimentata dalla pompa **P5**.

L' acqua fredda generata dal chiller a temperature tra 6 e 8 °C, viene inviata tramite la pompa **P4** al sistema di distribuzione della serra, in parallelo a quella prodotta dal chiller elettrico a compressione, montato in origine per raffrescare la serra.




Fig. 1, Schema Solar Cooling Serra F96

2-Sensori utilizzati

I misuratori impiegati sull' impianto consistono in:

1-n° 10 termoresistenze Pt 100 sulle linee di ingresso ed uscita di: Campo Solare, Acqua Calda, Acqua Refrigerata, Acqua di Torre e Caldaia.

2- n° 1 termistore NTC 2.2 K per l' aria esterna

3-n° 6 misuratori di portata ad induzione magnetica sulle correnti di: Campo Solare, Acqua Calda, Acqua Refrigerata, Acqua di Torre, Acqua all' Assorbitore e Caldaia.

2.1 Sensori PT 100, correnti liquide.

Per le correnti liquide interessate da scambio termico, è fondamentale avere una buona precisione sulla lettura del salto di temperatura che è direttamente proporzionale alla potenza termica di ciascuna apparecchiatura servita, per tale motivo sono state selezionate delle termoresistenze PT100.

La bassa resistenza di questi sensori rende la lettura maggiormente sensibile alla resistenza aggiuntiva dovuta alla lunghezza dei cavi di collegamento, per cui si è reso necessario il collegamento a quattro fili che nasce proprio per risolvere questo problema.

Le termoresistenze PT100 utilizzate sono del tipo per esterno della Sicest, reperite da fondi di magazzini dell' allora CNEN, verosimilmente produzione anni '60-70.

I sensori sono stati posizionati in appositi pozzetti termometrici riempiti di olio, all' ingresso e all' uscita di ciascuna sezione di impianto interessata a produzione o assorbimento di calore.

Dato il tipo di elemento sensibile (filo di platino) la loro durata è praticamente illimitata e sono governate da una legge molto lineare, almeno nel campo 0-100 °C di nostro interesse:

$$1) \quad R = R_0 + \alpha(T - T_0)$$

Dove R è la resistenza alla temperatura generica T (°C), R₀ (100 ohm) è la resistenza a T₀ (0°C) e α è la costante caratteristica del materiale del sensore, pari a 0.3850 ohm/°C (norma DIN IEC 751).

Già usando la relazione 1) tal quale, ci si aspetta una precisione tipica di +/- 0.3°C, tuttavia per minimizzare l' errore generato dalle tolleranze di produzione e quello introdotto dalla catena di misura, ogni sensore è stato successivamente calibrato sul campo mediante due letture ai valori fissi 0°C e 100 °C tramite la seguente relazione lineare (retta per due punti):

$$2) \quad T_c = 100 * T / (T_{100} - T_0) - T_0 * 100 / (T_{100} - T_0)$$

Dove T_c è la lettura corretta, T₁₀₀ e T₀ sono le letture non corrette ottenute tramite la 1) ai valori fissi reali di calibrazione.

Per la generazione dei valori fissi di calibrazione sono stati utilizzati i pozzetti di taratura mod. JOFRA 200 S (+20/+200°C) e D 50 RC (-50/+75°C), opportunamente verificati con termometri a mercurio di alta precisione.



Fig.2, Taratura sul campo dei sensori di temperatura PT100

Per i 10 sensori PT100 utilizzati sono state rilevate le seguenti letture per la calibrazione:

Sensore PT100	Posizione	T ₀	T ₁₀₀
0	Solare IN	-0.31	99.38
1	Solare OUT	-0.50	99.25
2	Acqua Calda IN	-0.17	99.90
3	Acqua Calda OUT	-0.30	99.84
4	Acqua Torre IN	-0.27	99.56
5	Acqua Torre OUT	-0.15	99.58
6	Acqua Fredda IN	-0.67	99.90
7	Acqua Fredda OUT	-0.34	99.82
8	Caldaia IN	0.02	99.84
9	Caldaia OUT	-0.42	99.90

Tabella 1: calibrazione sensori PT100

2.2 Sonda NTC, temperatura dell' aria esterna.

Per la temperatura dell' aria esterna è stata reperita una sonda NTC 2.2 K che è stata calibrata mediante il pozzetto di taratura JOFRA D 50 RC, vedi tabella 2.

T °C	OHM
-9.58	8896.3
0.45	5782.7
10.47	3823.6
20.5	2561.2
30.52	1752.9
40.54	1219
50.57	863.19

Tabella 2: calibrazione sonda NTC 2.2K

Le sonde NTC, molto usate nel settore automobilistico e degli elettrodomestici, sono caratterizzate da un costo molto contenuto e presentano il grosso vantaggio di avere una resistenza elevata, che minimizza l' errore generato dalla resistenza aggiuntiva dovuta alla lunghezza del filo elettrico utilizzato per la connessione.

Tali sonde, per contro, sono affette da una forte non linearità, tuttavia, interpolando i dati misurati della tabella 2 con la formula di Steinhart e Hart [3], si è ottenuta uno scarto massimo inferiore a di +/- 0.05 °C sulla lettura:

$$3) \quad T_k = 1 / (.00144060671799 + .000225574189225 * \ln(R) + 4.01258778741E-7 * (\ln(R))^3)$$

Dove R è la resistenza in ohm misurata alla temperatura Tk in °K.

2.3 Misuratori di flusso ad induzione magnetica, correnti liquide.

Per la misura delle portate sono stati utilizzati i misuratori ad induzione KHRONE OPTIFLUX 5100, originariamente installati sull' impianto.

Tali sensori offrono una precisione di lettura pari allo 0.1 % del valore letto e possono trasmettere la lettura tramite una uscita analogica lineare 4-20 mA.

Il fatto di utilizzare una uscita in corrente permette di non tenere conto della lunghezza del cavo elettrico di connessione.

Per la lettura finale, ai capi di uscita di ciascun misuratore è stata posizionata una resistenza da 250 ohm all' 1 % per trasformare il segnale di lettura nel campo lineare 1-5 V.

Con questa configurazione è possibile risalire dal voltaggio di lettura alla portata di ogni singolo flussimetro, con la seguente relazione lineare:

$$3) \quad F = F_s \cdot (V_{\text{lett}} - V_{\text{min}}) / (V_{\text{max}} - V_{\text{min}})$$

Dove, $F(L/h)$ è la portata letta, $F_s (L/h)$ è il fondo scala di ciascun flussimetro, V_{max} (5 volt) e V_{min} (1 volt) sono i limiti di lettura per il voltaggio e V_{lett} è il voltaggio effettivamente letto.

I fondo scala di ciascun sensore sono stati impostati in accordo alla tabella 3:

Corrente	Fondo Scala (L/h)
Solare	6000
Acqua Calda	5000
Acqua Torre	10000
Acqua Assorbitore	6000
Acqua Fredda	5000
Caldaia	2000

Tabella 3, fondo scala misuratori di portata

3- Acquisizione dati

Per acquisire i valori dei vari sensori è stato utilizzato un sistema composto da un PC IBM Intellistation M Pro, della fine degli anni '90, dotato di scheda GPIB National PCI, e da un data logger HP 3497 A [4], recuperato dal vecchio impianto CAPOC, dismesso intorno agli anni '90.

Il software di acquisizione è stato realizzato su misura utilizzando il BASIC HTBWIN 9.1 della Transera [6], capace di replicare su piattaforma Windows il vecchio linguaggio Rocky Mountain BASIC (BASIC 9000), in dotazione ai computer HP serie 200 degli anni '80, un tipo di BASIC avanzato con funzioni molto simili al FORTRAN .

Rispetto a software più moderni e complessi, come ad esempio il LabView della National Instruments, tale linguaggio di programmazione dal sapore vintage ma tuttora in produzione (versione 10.0), permette sia una semplice realizzazione di programmi di acquisizione dati sia una agevole ed elegante presentazione su schermo dei dati acquisiti, grazie alle librerie grafiche di cui è stato dotato rispetto al passato.

Il fatto di richiedere poca memoria RAM (32 Mb) per la sua esecuzione, permette di utilizzare PC anche obsoleti, come nel nostro caso, non rinunciando alle piene prestazioni ottenibili.

L' HP 3497 A, dotato di voltmetro da 5 e ½ digit e generatore di corrente (Opt. 001), permette di acquisire autonomamente sino a 20 misure analogiche miste in Volts e Ohm tramite ognuna delle 5 schede multiplexer a relais di cui è dotabile, e fino a un massimo di 10 schede in più, tramite l' apposito extender HP 3498 A, del quale se ne possono usare ben 13, per un totale di 135 schede, con complessivi 2700 canali di misura [5] .

Se si prevede di usare il collegamento a quattro fili per i sensori di temperatura resistivi, che richiede di usare due canali per ciascun sensore, il massimo numero di sensori per scheda scende a 10.

Esistevano all' epoca un totale di 11 tipi di schede diverse, dedicate a termocoppie, strain gauge, uscite in tensione e corrente, relais, I/O digitale, etc etc.

Tale apparecchio era realizzato in USA con costi e qualità molto elevati, tanto che a distanza di 30 anni e più, molti esemplari, come il nostro, ancora funzionano perfettamente.

In particolare sono stati utilizzati soltanto tre dei cinque slot disponibili per le schede di acquisizione dei segnali analogici.

Nella nostra configurazione, il data Logger HP 3497 A è dotato soltanto del voltmetro (Opt. 001) e di 3 schede multiplexer a relais (Opt. 010) HP44421A, adatte a leggere voltaggi e resistenze, posizionate negli slot 0, 1 e 2.

Lo **Slot 0** è stato destinato alla lettura delle dieci temperature da sonda Pt100 a quattro fili: *questa scelta comporta anche una operazione hardware* sulla scheda stessa con l'apertura di tre jumper dedicati, J4, J5 e j6, e l'alimentazione dell'apposita morsettiera "COM B" con una corrente calibrata di riferimento (1 mA) generata dal voltmetro (Opt. 001).




Fig. 3, collegamento sensori resistivi a quattro fili

Sensore PT100	Posizione	canali
0	Solare IN	0,10
1	Solare OUT	1,11
2	Acqua Calda IN	2,12
3	Acqua Calda OUT	3,13
4	Acqua Torre IN	4,14
5	Acqua Torre OUT	5,15
6	Acqua Fredda IN	6,16
7	Acqua Fredda OUT	7,17
8	Caldaia IN	8,18
9	Caldaia OUT	9,19

Tabella 4, collegamenti Slot 0

Lo **Slot 1** è stato utilizzato solo in parte per leggere i voltaggi generati dai 6 misuratori di portata ad induzione magnetica per il tramite delle altrettante resistenze da 250 ohm poste in parallelo: rimangono pertanto altri 14 canali liberi per eventuali successivi ampliamenti.

Flussimetro	canale
Solare	0
Acqua Calda	1
Acqua Torre	2
Acqua Assorbitore	3
Acqua Fredda	4
Caldaia	5

Tabella 5, collegamenti Slot 1




Fig. 4, collegamento per lettura voltaggi DC

Lo **Slot 2** è per ora impiegato per leggere soltanto il termistore NTC dell' aria esterna con collegamento a due fili sul canale 0: rimangono pertanto a disposizione altri 19 canali per altrettanti sensori resistivi con collegamento a due fili.

Anche qui si è resa necessaria l' alimentazione della morsetteria "COM B" della scheda con l' amperaggio di riferimento da 1 mA, ma non l' apertura dei jumper dedicati, dato che la lettura della resistenza delle sonde NTC è sufficiente a soli due fili.




Fig. 5, collegamento sensori resistivi a due fili

La lettura finale effettuata dal modulo voltmetrico di precisione del data logger è del tipo “*guarded*”, pertanto per ogni canale la calza dei cavi di collegamento ai sensori deve essere collegata sul sensore all’ uscita -, mentre sulla scheda va collegata alla morsettiera “G” .

Sempre sulla scheda alle morsettiere “H” vanno collegati i segnali + e a quelle “L” i segnali -, per ciascun canale.

3.1 Programma di acquisizione dati

Le grandezze primitive acquisite tramite il data logger, valori in ohm e in volt, tradotte in temperature e portate mediante le Subroutines “**Rtd**”, “**Ntc2.2k**” e “**Port**”, sono state successivamente manipolate per generare altre *grandezze derivate*, quali potenze termiche e COP del chiller ad assorbimento.

In particolare le *potenze termiche* in gioco in ciascuna apparecchiatura sono state calcolate tramite lo stesso programma di acquisizione dati mediante la seguente formula:

$$5) \quad Q = F * C_p * R_o * (T_{out} - T_{in}) / 860$$

Dove Q (Kw) è la potenza termica, F (L/h) è la portata liquida, Cp (Kcal/Kg °C) e Ro (Kg/m³) sono il calore specifico e la densità della corrente calcolati alla temperatura media, Tout e Tin (°C) sono le temperature di uscita ed ingresso dall’ apparecchiatura considerata: Solare, Acqua Calda, Acqua di Torre, Acqua Fredda e Caldaia.

Cp e Ro tengono conto anche dell’ eventuale presenza di Glicole etilenico nella linea dell’ acqua fredda (unica linea dove viene utilizzato), tramite una concentrazione % in peso inserita all’ interno del programma una volta per tutte.

Per il calcolo di Cp e Ro viene utilizzata la Sub Routine “**Termof**” che contiene alcuni polinomi interpolatori delle due grandezze, in funzione di temperatura e percentuale di glicole in peso, nel campo 0–120 °C e 0-60% in peso di glicole etilenico

La *portata al condensatore* è stata ottenuta semplicemente sottraendo alla portata completa di torre quella dell’ assorbitore.

Il *COP* del chiller ad assorbimento è stato calcolato come rapporto tra la potenza termica all’ evaporatore (Acqua Fredda) e quella al generatore (Acqua Calda), precedentemente calcolate.

Preventivamente è stata generata a parte, con l’ applicazione per windows “**scr_bld.exe**”, una pagina grafica composta da svariati *Widgets* per la presentazione dei dati numerici di grandezze primitive e derivate, da moduli di limite per le grandezze critiche e da un grafico mobile.

Tale pagina grafica, grazie all’ opzione “**BPLUS**” che permette l’ uso delle librerie grafiche aggiuntive, viene caricata automaticamente dal programma di acquisizione dati per essere popolata dai valori delle grandezze acquisite di volta in volta, vedi fig. 4.

Sono presenti quattro gruppi di dati alfanumerici: il primo per le 10 temperature di processo, il secondo per le portate, il terzo per le prestazioni e l’ ultimo dedicato agli extra.

Sono poi presenti cinque campi, corredati di limiti di normalità inferiore e superiore, per il corretto funzionamento del chiller ad assorbimento Yazaki SC5 e del campo Solare, sono cinque: portate Acqua Solare, Acqua Calda e Acqua di Torre, temperature uscita Solare ed ingresso Acqua Calda.

Grandezza	Limite inf.	Limite sup.
Portata Solare	4500	5500
Portata Acqua Calda	4000	4400
Portata Torre	7500	9100
T. Solare OUT	75	100
T. Acqua Calda IN	75	95

Tabella 6, limiti critici di funzionamento

Per ultimo è stato inserito un grafico mobile su cui si è scelto di visualizzare gli andamenti delle 10 temperature di processo più quella dell' aria esterna al variare del tempo.




Fig. 6, Schermata di presentazione dei dati sperimentali

Il programma genera automaticamente un file di testo dove memorizzare i dati acquisiti, ed aggiunge ad ogni campionamento una stringa contenente tutti i valori delle variabili primitive e derivate più l' ora del campionamento stesso, con valori separati da virgole (CSV).

Il nome di ciascun file, da 500 campionamenti al massimo (modificabile), è composto dalla data e dall' ordine giornaliero. (Es. : 1Jul2016_0.txt ; 1Jul2016_1.txt ; 1Jul2016_2.txt).

Per una rapida identificazione, la prima fila di ogni file contiene le intestazioni e l' unità di misura di ogni grandezza memorizzata, per un totale di 501 righe e 25 colonne.

Tali file di testo, piccoli e maneggevoli (~80 KB) possono essere immediatamente utilizzati esternamente tramite Excel o programmi similari.

Per fare partire l' acquisizione dati bisogna prima eseguire il file "HTBwin.exe", poi caricare il file "SAD F96_5.bas" (versione attuale) ed infine premere il tasto "RUN".

A questo punto il programma chiede il tempo di ogni campionamento in secondi, l' ampiezza del grafico mobile in ore e quindi, se il data logger è acceso, inizia immediatamente ad acquisire i dati.

Cliccando sull' *icona a forma di rettangolino grigio* in alto a sinistra della schermata grafica e rispondendo opportunamente tramite la tastiera, è possibile scegliere in qualunque momento, anche a programma avviato, l' ampiezza in ore della finestra mobile temporale del grafico insieme alla possibilità di fermare l' acquisizione dati,

I file di testo generati vengono salvati in un Hard Disk a parte, sotto la cartella "DATI SERRA".

Conclusioni

Il nuovo sistema di acquisizione dati per la serra a contenimento è stato realizzato a costo zero e messo in funzione per l' estate in corso.

La visualizzazione dei dati è stata scelta per capire con uno sguardo se il funzionamento dell' impianto di solar cooling della serra è ottimale, ma anche in vista delle frequenti visite esterne.

Il sistema adottato permetterà in futuro di aggiungere qualunque altra variabile di interesse per la serra: irraggiamento solare, temperature dei moduli, umidità, portate di acqua agli aerotermini interni, grandezze chimico/fisiche etc. etc. praticamente senza limite di numerosità.

Ringraziamenti:

Si ringraziano i colleghi: Antonio Scotini per l' aiuto ricevuto nel ripristinare i condensatori della mother board del PC usato; Luca Nardi per l' aiuto nel posizionare cavi, sensori e l' intero hardware.

Bibliografia:

- [1] G. Corallo, A. Simonetti, Ricerca di Sistema nel Settore Elettrico, Realizzazione di un impianto di solar cooling a servizio di una serra per colture intensive, Report RdS/2012/129
- [2] G. Corallo, A. Simonetti, Ricerca di Sistema nel Settore Elettrico, Gli impianti di Solar Cooling al servizio di serre agricole: prove sperimentali, Report RDS/2013/89
- [3] John S. Steinhart, Stanley R. Hart, Calibration curves for thermistors, in Deep Sea Research, vol. 15, 1968, pp. pp. 497-503
- [4] HP JOURNAL, July 1981, vol.32, n. 7
- [5] Hewlett Packard, Model 3497A; Operating, Programming and Configuration Manual, Dec. 1982
- [6] <http://transera.com/htbasic/>

Appendice A: listato del programma

1 ! PROGRAMMATO DA G. CORALLO LUGLIO 2016

2 !

3 ! SISTEMA ACQUISIZIONE DATI SOLAR COOLING F96

4 !

5 !

10 LOAD BIN "BPLUS"

11 OPTION BASE 0

12 DIM S\$(0:9)[30],T\$(0:9)[30],F\$(0:6)[30],Q\$(0:5)[30],Ex\$(0:5)[30],T0(0:9),T100(0:9),Ex(0:5)

13 DIM T(0:10),F(0:6),F0(0:6),Q(0:5),Tm(0:4),G(0:23),G\$(0:23)[30],Ff\$(11),Sm\$(0:1)[30]

14 !

15 ! STRINGHE CAMPI DISPLAY

16 !

17 T\$(0)="SOLARE IN °C"

18 T\$(1)="SOLARE OUT °C"

19 T\$(2)="ACQUA CALDA IN °C"

20 T\$(3)="ACQUA CALDA OUT °C"

21 T\$(4)="TORRE IN °C"

22 T\$(5)="TORRE OUT °C"

23 T\$(6)="ACQUA FREDDA IN °C"

25 T\$(7)="ACQUA FREDDA OUT °C"

26 T\$(8)="CALDAIA IN °C"

27 T\$(9)="CALDAIA OUT °C"

28 !

29 F\$(0)="SOLARE (L/h)"

30 F\$(1)="ACQUA CALDA (L/h)"

31 F\$(2)="TORRE (L/h)"

32 F\$(3)="ASSORBITORE (L/h)"

33 F\$(4)="CONDENSATORE (L/h)"

34 F\$(5)="ACQUA FREDDA (L/h)"

35 F\$(6)="CALDAIA (L/h)"

47 !

57 Q\$(0)="SOLARE (KW)"

58 Q\$(1)="ACQUA CALDA (KW)"

59 Q\$(2)="ACQUA TORRE (KW)"

60 Q\$(3)="ACQUA FREDDA (KW)"

62 Q\$(4)="CALDAIA (KW)"

63 Q\$(5)="COP"

64 !

65 Ex\$(0)="ARIA ESTERNA (°C)"

66 !

67 ! CHECK DATALOGGER

```

68 !
69 Scan=709
70 CLEAR Scan
71 OUTPUT Scan;"ST1"
72 ENTER Scan;Rispl
73 ! DISP Rispl
74 !PAUSE
75 IF Rispl<.8.E+8 THEN
76 CLEAR SCREEN
77 PRINT "ATTENZIONE IL DATA LOGGER E' GUASTO"
78 STOP
79 END IF
80 !
81 ! SET DATALOGGER
82 !
83 OUTPUT Scan;"ST0"
84 OUTPUT Scan;"VC3" ! 1 mA
85 OUTPUT Scan;"VA1" ! AUTOZERO ON
86 OUTPUT Scan;"VR5" ! AUTORANGE
87 !
88 !
89 INPUT "TEMPO DI CAMPIONAMENTO (sec)=?",Tcamp
91 INPUT "AMPIEZZA GRAFICO (ORE)=?",Span
250 Day0=VAL(DATE$(TIMEDATE))
251 Tempo0=TIMEDATE
252 !
253 ASSIGN @Sys TO WIDGET "SYSTEM"
254 CONTROL @Sys;SET ("*LOAD": "F96.SCR")!LEGGE LO SCHERMO CREATO CON SCREEN BUILDER
255 !
256 !
257 STATUS @Sys;RETURN ("*WIDGETS":Num_widgets)!LEGGE IL NUMERO DI WIDGETS PRESENTI
258 PRINT "NUMERO DI WIDGETS=";Num_widgets
259 !
260 !
261 !
262 Sm$(0)="STOP ACQUISIZIONE"
263 Sm$(1)="FONDO SCALA GRAFICO (ORE)"
264 !
265 CONTROL @Sys;SET ("*NAME": "Panel0", "VISIBLE":1, "SYSTEM MENU":Sm$(*))
266 CONTROL @Sys;SET ("*NAME": "Panel0", "TITLE": "SAD SOLAR COOLING SERRA F96 - V. 1.5 G. Corallo SSPT-STs") !CAMBIA IL VALORE DEL
TITOLO
267 CONTROL @Sys;SET ("*NAME": "Panel0/String0", "BACKGROUND":1, "VALUE": " TEMPERATURE") !CAMBIA IL VALORE DELLA STRINGA
268 CONTROL @Sys;SET ("*NAME": "Panel0/String1", "BACKGROUND":1, "VALUE": " PORTATE")
269 CONTROL @Sys;SET ("*NAME": "Panel0/String2", "BACKGROUND":1, "VALUE": " PRESTAZIONI")

```

```

270 CONTROL @Sys;SET ("*NAME": "Panel0/String26", "BACKGROUND":1, "VALUE": "  EXTRA")
271 !
272 ON EVENT @Sys, "SYSTEM MENU" GOSUB Scelta! SYSTEM MENU
281 !
282 ! TEMPERATURE
283 !
284 FOR J=0 TO 9
285 IF J=0 OR J=1 THEN
286 Penna=3! GIALLO
287 END IF
288 IF J=4 OR J=5 THEN
289 Penna=4! VERDE
290 END IF
291 IF J=2 OR J=3 THEN
292 Penna=2! ROSSO
293 END IF
294 IF J=6 OR J=7 THEN
295 Penna=5! CIANO
296 END IF
297 IF J=8 OR J=9 THEN
298 Penna=7! MAGENTA
299 END IF
300 !
301 S$(J)="String"&VAL$(J+3)
302 !PRINT S$(J),T$(J)
303 CONTROL @Sys;SET ("*NAME": "Panel0/"&S$(J), "PEN":Penna, "BACKGROUND":0, "VALUE":T$(J))
304 NEXT J
305 !
306 ! PORTATE
307 !
308 FOR J=0 TO 6
309 IF J=0 THEN
310 Penna=3! GIALLO
311 END IF
312 IF J=1 THEN
313 Penna=2! ROSSO
314 END IF
315 IF J=2 THEN
316 Penna=4! VERDE
317 END IF
318 IF J=3 THEN
319 Penna=4! VERDE
320 END IF

```

```

321 IF J=4 THEN
322 Penna=4 ! VERDE
323 END IF
324 IF J=5 THEN
325 Penna=5 ! CIANO
326 END IF
327 IF J=6 THEN
328 Penna=7 ! MAGENTA
329 END IF
330 !
331 S$(J)="String"&VAL$(J+13)
332 CONTROL @Sys;SET ("*NAME":"Panel0"&S$(J),"PEN":Penna,"BACKGROUND":0,"VALUE":F$(J))
333 NEXT J
334 !
335 ! PRESTAZIONI
336 !
337 FOR J=0 TO 5
338 IF J=0 THEN
339 Penna=3 ! GIALLO
340 END IF
341 IF J=1 THEN
342 Penna=2 ! ROSSO
343 END IF
344 IF J=2 THEN
345 Penna=4 ! VERDE
346 END IF
347 IF J=3 THEN
348 Penna=5 ! CIANO
349 END IF
350 IF J=4 THEN
351 Penna=7 ! MAGENTA
352 END IF
353 IF J=5 THEN
354 Penna=1 ! BIANCO
355 END IF
356 S$(J)="String"&VAL$(J+20)
357 CONTROL @Sys;SET ("*NAME":"Panel0"&S$(J),"PEN":Penna,"BACKGROUND":0,"VALUE":Q$(J))
358 NEXT J
360 !
361 ! EXTRA
362 !
363 CONTROL @Sys;SET ("*NAME":"Panel0/String27","PEN":6,"BACKGROUND":1,"VALUE":Ex$(0)) ! ARIA ESTERNA: COLORE BLU; SFONDO BIANCO
364 !

```

```

365 !
366 ! GRAFICO
367 !
368 CONTROL @Sys;SET ("*NAME": "Panel0/Stripchart0", "TRACE COUNT": 11, "MINIMUM SCROLL": 90, "SHARED X": 1)
369 !
370 ! Set X-axis and Y-axis origin, range, and labeling
371 !
372 Span=3600*Span
373 !
374 CONTROL @Sys;SET ("*NAME": "Panel0/Stripchart0", "CURRENT AXIS": "Y")
375 CONTROL @Sys;SET ("*NAME": "Panel0/Stripchart0", "ORIGIN": -5, "RANGE": 110, "AXIS LABEL": "C")
376 CONTROL @Sys;SET ("*NAME": "Panel0/Stripchart0", "CURRENT AXIS": "X")
377 CONTROL @Sys;SET ("*NAME": "Panel0/Stripchart0", "ORIGIN": 0, "RANGE": Span, "AXIS LABEL": "TEMPO", "NUMBER FORMAT": "HOURS", "NUMBER
FORMAT": "CLOCK24")
378 CONTROL @Sys;SET ("*NAME": "Panel0/Stripchart0", "PEN": 3, "BACKGROUND": 0, "GRID PEN": 0, "SHOW GRID": "MAJOR") ! ASSI GIALLI, SFONDO NERO, GRIGLIA
NERA
379 !
380 !
381 !
382 !
383 FOR J=0 TO 9
384 !
385 IF J=0 OR J=1 THEN
386 Penna=3 ! GIALLO
387 END IF
388 IF J=4 OR J=5 THEN
389 Penna=4 ! VERDE
390 END IF
391 IF J=2 OR J=3 THEN
392 Penna=2 ! ROSSO
393 END IF
394 IF J=6 OR J=7 THEN
395 Penna=5 ! CIANO
396 END IF
397 IF J=8 OR J=9 THEN
398 Penna=7 ! MAGENTA
399 END IF
400 !
401 CONTROL @Sys;SET ("*NAME": "Panel0/Stripchart0", "CURRENT TRACE": J+1, "TRACE PEN": Penna, "TRACE LABEL": T$(J), "VALUE": T(J))
402 !
403 NEXT J
404 !
405 Penna=6 ! BLU
406 CONTROL @Sys;SET ("*NAME": "Panel0/Stripchart0", "CURRENT TRACE": 11, "TRACE PEN": Penna, "TRACE LABEL": Ex$(0), "VALUE": T(10))

```

407 !
409 !
410 Camp=1
411 Rip=0
412 !
413 !
422 !
423 ! DATI CALIBRAZIONE PT100 SLOT 0
424 !
425 T0(0)=-.31 !LETTURA NON CORRETTA A 0°C EFFETTIVI
426 T100(0)=99.38 !LETTURA NON CORRETTA A 100°C EFFETTIVI
427 T0(1)=-.50
428 T100(1)=99.25
429 T0(2)=-.17
430 T100(2)=99.9
431 T0(3)=-.3
432 T100(3)=99.84
433 T0(4)=-.27
434 T100(4)=99.56
435 T0(5)=-.15
436 T100(5)=99.58
437 T0(6)=-.67
438 T100(6)=99.90
439 T0(7)=-.34
440 T100(7)=99.82
441 T0(8)=.02
442 T100(8)=99.84
443 T0(9)=-.42
444 T100(9)=99.9
445 !
446 !
447 ! FONDO SCALA MISURATORI DI PORTATA
448 !
449 Fs(0)=6000 ! SOLARE (L/h)
450 Fs(1)=5000 ! ACQUA CALDA (L/h)
451 Fs(2)=10000 ! ACQUA TORRE (L/h)
452 Fs(3)=5000 ! ACQUA REFRIGERATA (L/h)
453 Fs(4)=6000 ! ACQUA ASSORBITORE (L/h)
454 Fs(5)=2000 ! ACQUA CALDAIA (L/h)
455 !
456 LOOP ! INIZIO ACQUISIZIONE
457 !
458 ! INGRESSO SEGNALI

```

459 !
460 ! GRANDEZZE PRIMITIVE
461 !
462 !
463 Slot=0 ! TERMORESISTENZE 4 FILI (CANALI ORDINATI SECONDO STRINGHE DISPLAY)
466 !
467 FOR Ncan=0 TO 9
468 Ncan$=VAL$(Slot*20+Ncan)
469 Ncan2$=VAL$(10+Ncan)
470 !PRINT "AC";Ncan$;";";Ncan2$
471 ! PAUSE
472 OUTPUT Scan;"AC";Ncan$;";";Ncan2$
473 ENTER Scan;V(Ncan)
474 R(Ncan)=V(Ncan)/.001
475 CALL Rtd(R(Ncan),T0(Ncan),T100(Ncan),T(Ncan))
476 IF T(Ncan)>200 THEN
477 T(Ncan)=200
478 END IF
479 ! PRINT "V(";Ncan;")=";V(Ncan),"R(";Ncan;")=";R(Ncan),"T(";Ncan;")=";T(Ncan),T$(Ncan)
480 !PRINT USING "AA,DD,AA.DDDDD.D,XX,K";"T(";Ncan;")=";T(Ncan),T$(Ncan)
481 !
482 NEXT Ncan
483 !PRINT
484 !PRINT
485 !
486 Slot=1 ! MISURATORI DI PORTATA 4-20 mA + 250 ohm parallelo (CANALI ORDINATI SECONDO FONDO SCALA)
487 !
488 FOR Ncan=0 TO 5
489 Ncan$=VAL$(Slot*20+Ncan)
490 !PRINT "AC";Ncan$
491 ! PAUSE
492 OUTPUT Scan;"AC";Ncan$
493 ENTER Scan;V(Ncan)
494 CALL Port(V(Ncan),Fs(Ncan),F0(Ncan))
495 !
496 ! RIORDINA LE PORTATE PER LA PRESENTAZIONE GRAFICA (CANALI NON CORRISPONDONO AD ORDINE STRINGHE DISPLAY)
497 !
498 IF Ncan<=2 THEN
499 F(Ncan)=F0(Ncan)
500 END IF
501 IF Ncan=3 THEN
502 F(5)=F0(Ncan)
503 END IF

```

```

504 IF Ncan=4 THEN
505 F(3)=F0(Ncan)
506 END IF
507 IF Ncan=5 THEN
508 F(6)=F0(Ncan)
509 END IF
510 !
511 !PRINT "V(";Ncan;")=";V(Ncan),"F(";Ncan;")=";F(Ncan),F$(Ncan)
512 !PRINT USING "AA,DD,AA,DDDD.D,XX,K";"F(";Ncan;")=";F(Ncan)
513 !PRINT
514 NEXT Ncan
515 ! PRINT
516 ! PRINT
517 !
518 Slot=2 ! EXTRA
519 Ncan=0
520 Ncan$=VAL$(Slot*20+Ncan)
521 OUTPUT Scan;"AC";Ncan$ !! TERMISTORE 2.2K 2 FILI PER L' ARIA ESTERNA
522 ENTER Scan;V2(Ncan)
523 R2(Ncan)=V2(Ncan)/.001
524 CALL Ntc2_2k(R2(Ncan),T2(Ncan))
525 Ex(0)=T2(Ncan)
526 T(10)=Ex(0)
527 !
528 ! GRANDEZZE DERIVATE
529 !
530 !
531 F(4)=F(2)-F(3) ! PORTATA AL CONDENSATORE
532 !
533 I=0
534 FOR J=0 TO 8 STEP 2
535 IF I=3 THEN
536 Gliw=0      ! % IN PESO DI GLICOLE
537 ELSE
538 Gliw=0
539 END IF
540 Tm(I)=(T(J)+T(J+1))/2    ! T media °C
541 CALL Termof(Gliw,Tm(I),Cp(I),Ro(I))
542 IF I>=3 THEN
543 Ii=I+2
544 ELSE
545 Ii=I
546 END IF

```

```

547 Q(I)=F(I)*Cp(I)*Ro(I)*(T(J+1)-T(J))/860      ! KW
548 !PRINT USING "AA,DD,AA,DD.D,XX,K";"Q(";I;")=";Q(I),Q$(I)
549 I=I+1
550 NEXT J
551 !PRINT
552 !PRINT
553      !
554 Cop=Q(3)/(Q(1)+.0001)
555 Q(5)=Cop
556      !
557 !PRINT USING "K,D.DD";"COP=";Cop
558 !PRINT
559 !PRINT
560 !PRINT
561 !PRINT
562 !
578 ! GRANDEZZE DA ARCHIVIARE, STRINGHE CON VIRGOLA FINALE E NUMERI ARROTONDATI ALLA PRIMA CIFRA DECIMALE SALVO COP ALLA SECONDA
579 !
580 G0$="ORARIO"&","
581 Ora$=TIMES$(TIMEDATE)&","
582 FOR J=0 TO 9
583 G(J)=INT(T(J))+DROUND(FRACT(T(J)),1)
584 G$(J)=T$(J)&","
585 NEXT J
586 FOR J=0 TO 6
587 G(J+10)=INT(F(J))+DROUND(FRACT(F(J)),1)
588 G$(J+10)=F$(J)&","
589 NEXT J
590 FOR J=0 TO 5
591 IF J=5 THEN
592 G(J+17)=INT(Q(J))+DROUND(FRACT(Q(J)),2)
593 G$(J+17)=Q$(J)&","
594 ELSE
595 G(J+17)=INT(Q(J))+DROUND(FRACT(Q(J)),1)
596 G$(J+17)=Q$(J)&","
597 END IF
598 NEXT J
599 G$(23)=Ex$(0)
600 G(23)=INT(Ex(0))+DROUND(FRACT(Ex(0)),1)
601 !
602 !
603 !
604 FOR J=0 TO 9

```

```

605 S$(J)="Number"&VAL$(J)
606 CONTROL @Sys;SET ("*NAME":"Panel0/"&S$(J),"REAL NOTATION":"FIXED","REAL RESOLUTION":1,"VALUE":T(J))
607 NEXT J
608 !
609 FOR J=0 TO 6
610 S$(J)="Number"&VAL$(J+10)
611 CONTROL @Sys;SET ("*NAME":"Panel0/"&S$(J),"REAL NOTATION":"FIXED","REAL RESOLUTION":0,"VALUE":F(J))
612 NEXT J
613 !
614 FOR J=0 TO 5
615 S$(J)="Number"&VAL$(J+17)
616 IF J=5 THEN
617 CONTROL @Sys;SET ("*NAME":"Panel0/"&S$(J),"REAL NOTATION":"FIXED","REAL RESOLUTION":2,"VALUE":Q(J))
618 ELSE
619 CONTROL @Sys;SET ("*NAME":"Panel0/"&S$(J),"REAL NOTATION":"FIXED","REAL RESOLUTION":1,"VALUE":Q(J))
620 END IF
621 NEXT J
622 !
623 S$(0)="Number"&VAL$(0+23)
624 CONTROL @Sys;SET ("*NAME":"Panel0/"&S$(0),"REAL NOTATION":"FIXED","REAL RESOLUTION":1,"VALUE":Ex(0))
625 !
626 ! LIMITI
627 !
628 !
629 CONTROL @Sys;SET ("*NAME":"Panel0/Limits0","MINIMUM":4000,"MAXIMUM":6000,"LOW LIMIT":4500,"HIGH LIMIT":5000,"VALUE":F(0))
630 !
631 CONTROL @Sys;SET ("*NAME":"Panel0/Limits1","MINIMUM":3500,"MAXIMUM":5000,"LOW LIMIT":4000,"HIGH LIMIT":4400,"VALUE":F(1))
632 !
633 CONTROL @Sys;SET ("*NAME":"Panel0/Limits2","MINIMUM":7000,"MAXIMUM":10000,"LOW LIMIT":7500,"HIGH LIMIT":9100,"VALUE":F(2))
634 !
635 CONTROL @Sys;SET ("*NAME":"Panel0/Limits3","MINIMUM":70,"MAXIMUM":110,"LOW LIMIT":75,"HIGH LIMIT":100,"VALUE":T(1))
636 !
637 CONTROL @Sys;SET ("*NAME":"Panel0/Limits4","MINIMUM":70,"MAXIMUM":100,"LOW LIMIT":75,"HIGH LIMIT":95,"VALUE":T(2))
638 !
639 ! GRAFICO
640 !
641 !
642 N=TIMEDATE
643 CONTROL @Sys;SET ("*NAME":"Panel0/Stripchart0","POINT LOCATION":N,"VALUES":T(*))
644 WAIT Tcamp
645 !
646 !
647 ! MEMORIZZAZIONE

```

```

648 !
649 Day=VAL(DATES(TIMEDATE))
650 IF Day<>Day0 THEN
651 Rip=0
652 Day0=Day
653 END IF
654 MASS STORAGE IS "F:\DATI SERRA"
655 Rip$=VAL$(Rip)
656 ON ERROR GOTO 684
657 F1$=DATE$(TIMEDATE)
658 F1$=F1$[1,2]
659 F2$=F1$[4,6]
660 F3$=F1$[10,11]
661 File$=F1$&F2$&F3$&CHR$(95)&Rip$&".txt"
662 !PRINT File$
663 !PAUSE
664 !
665 Campmax=500 ! CAMPIONAMENTI MAX PER FILE
666 IF Camp=1 THEN
667 CREATE File$,0
668 ASSIGN @File TO File$;FORMAT ON
669 !
670 OUTPUT @File USING "K";G0$,G$(*)
671 !
672 END IF
673 !
674 OUTPUT @File;Ora$;G$(*)
675 !
676 IF Camp=Campmax THEN
678 Camp=1
679 ASSIGN @File TO *
680 GOTO 689
681 END IF
682 !
683 GOTO 688
684 OFF ERROR
685 Rip=Rip+1
686 GOTO 655
687 !
688 Camp=Camp+1
689 !
690 END LOOP
691 !

```

```

692 Scelta:! SYSTEM MENU
693 P$="Panel0"
694 STATUS @Sys;RETURN ("*NAME":P$,"SYSTEM MENU EVENT":Opzione)
695 SELECT Opzione
696 CASE 0
697 DIALOG "QUESTION","VUOI FERMARE ?","Boh
698 IF Boh=0 THEN
699 GOTO Finis
700 END IF
701 CASE 1
702 P$="FONDO SCALA (ORE)?"
703 DIALOG "NUMBER",P$,Btn;SET ("TITLE":"GRAFICO","FORMAT":"SHORT INTEGER"),RETURN ("VALUE":Span1)
704 Span1=Span1*3600
705 CONTROL @Sys;SET ("*NAME":"Panel0/Stripchart0","RANGE":Span1)
706 END SELECT
707 RETURN
708 !
709 Finis:!
710 ASSIGN @File TO *
711 ASSIGN @Sys TO *
712 CLEAR SCREEN
713 !
714 END
715 SUB Rtd(Ohm,T0,T100,Tc) !Pt100
716 T=(Ohm-100)/(100*3.85E-3) ! PT100 STANDARD
717 Tc=100*T/(T100-T0)-T0*100/(T100-T0) ! CORREZIONE LINEARE
718 IF Tc<0 OR Tc>100 THEN
719 DISP "TEMPERATURA FUORI SCALA (0 / +100)"
720 END IF
721 SUBEND
722 SUB Port(Volt,Fs,Lh)
723 Vmax=5
724 Vmin=1
725 Lh=Fs/(Vmax-Vmin)*(Volt-Vmin)
726 SUBEND
727 SUB Termof(Gliw,Tm,Cp,Ro)
728 !
729 !% GLICOLE IN PESO 10-60
730 !
731 IF Gliw<10 THEN
732 Gliw0=Gliw
733 Gliw=10
734 Flag$="A"

```

```

735 ELSE
736 Flag$="B"
737  !
738 END IF
739 A1=1005.781682
740 A2=1.260381493
741 A3=.000225223
742 A4=-.221703372
743 A5=-.001872923
744 A6=-.004211509
745 B1=4.310750439
746 B2=-.021074937
747 B3=-1.36201E-5
748 B4=-.000453536
749 B5=-1.91823E-6
750 B6=9.83781E-5
751 T=Tm
752 Ro=A1+A2*Gliw+A3*Gliw^2+A4*T+A5*T^2+A6*Gliw*T
753 Cp=B1+B2*Gliw+B3*Gliw^2+B4*T+B5*T^2+B6*Gliw*T
754 IF Flag$="A" THEN
755 Ro10=Ro
756 Cp10=Cp
757 Ro0=1001.23-.076727272727*T-.037034965035
758 Cp0=4.19703-.000874620999998*T+1.15029272727E-5*T^2
759 Ro=Ro0+(Ro10-Ro0)/10*Gliw0
760 Cp=Cp0+(Cp10-Cp0)/10*Gliw0
761 END IF
762 Cp=Cp/4.186  ! Kcal/Kg°C
763 Ro=Ro/1000  ! Kg/L
764 !
765 SUBEND
766 SUB Ntc2_2k(R,T) ! STEINHART & HART
769 Tk=1/(.00144060671799+.000225574189225*LOG(R)+4.01258778741E-7*(LOG(R))^3) ! STEINHART & HART
770 T=Tk-273.15
772 SUBEND

```

ENEA
Servizio Promozione e Comunicazione
www.enea.it

Stampa: Laboratorio Tecnografico ENEA - C.R. Frascati
ottobre 2016